

On souhaite gérer le score d'un match de rugby (c'est de saison) en temps réel. L'objectif est de faire la somme des points marqués au fur et à mesure du déroulement de la partie. Concevoir un programme VB pour atteindre cet objectif. Cette application doit comprendre au minimum les composants suivants :

- dans un souci esthétique, le logo de l'International Board, téléchargeable à l'adresse ftp://secco-nte.univ-lyon2.fr/darmont/DEUG/irb_logo.gif (cliquer à droite sur l'image et enregistrer sous) — utiliser un composant *Image* et sa propriété *Picture* ;
- deux zones de texte (*TextBox*) de nom **Eq1** et **Eq2** pour saisir le nom des équipes ;
- deux boutons d'option (*OptionButton*) de nom **PtEq1** et **PtEq2** permettant de déterminer quelle équipe marque des points ;
- une liste combinée (*ComboBox*) de nom **NbPt** permettant de sélectionner le nombre de points marqués à un moment donné (Transformation – 2 points, Pénalité – 3 points, Drop – 3 points, Essai – 5 points) ;
- un bouton de commande (*CommandButton*) de nom **Ajout** permettant d'ajouter à l'équipe indiquée par les boutons d'option les points indiqués par la liste combinée **NbPt** ;
- un bouton de commande (*CommandButton*) de nom **Res** permettant d'afficher le score dans une boîte de message ;
- un bouton de commande (*CommandButton*) de nom **Quit** permettant de quitter l'application.


Indications :

- Les scores de chaque équipe doivent être accessibles depuis les méthodes de plusieurs composants (procédures). Ils doivent donc être déclarés comme des variables globales dans la partie (Général) (Déclarations) du code.
- La liste combinée **NbPt** doit être de style 2 – *Dropdown List* (propriété *Style*). Dans la propriété *List* de **NbPt**, saisir les valeurs possibles (sous forme de chaînes).
- Dans la méthode *Ajout_Click*, déterminer d'abord le nombre de points (entier) à ajouter (avec un *select case*) selon les valeurs de la liste combinée **NbPt** (propriété *Text*), puis ajouter ces points au score idoine.
- Pour afficher une boîte de message, utiliser la procédure prédéfinie *MsgBox*. Ex. `call MsgBox("Un message", , "Titre")`. Pour afficher plusieurs chaînes et nombres, utiliser l'opérateur de concaténation `&` et la fonction `CStr()`.
- Pour quitter le programme, utiliser le mot clé *End*.

Questions subsidiaires :

- Ajouter un bouton de commande (*CommandButton*) de nom **Raz** permettant de remettre à zéro le nom des équipes et les scores.
- Ajouter deux listes (*ListBox*) de nom **List1** et **List2** permettant de garder une trace des points marqués par chaque équipe. Modifier la méthode *Raz_Click* pour également remettre à zéro le contenu de ces listes (utiliser la méthode *Clear* de **List1** et **List2**).

Apparence de l'application :



Code :

```
' Déclaration des scores totaux (variables globales)
Dim Score1 As Integer
Dim Score2 As Integer

Private Sub Ajout_Click()
    ' Déclaration nombre de points à ajouter
    Dim n As Integer
    ' Détermination de n selon la valeur de la combo box NbPt
    Select Case NbPt.Text
        Case "Transformation (2)"
            n = 2
        Case "Pénalité (3)"
            n = 3
        Case "Drop (3)"
            n = 3
        Case "Essai (5)"
            n = 5
    End Select
    ' Ajout de n au bon score selon l'équipe sélectionnée
    ' Ajout de "l'action marquée" dans la liste de l'équipe
    If PtEq1.Value Then
        Score1 = Score1 + n
        List1.AddItem (NbPt.Text)
    Else
        Score2 = Score2 + n
        List2.AddItem (NbPt.Text)
    End If
End Sub

Private Sub Quit_Click()
    ' Sortie du programme
    End
End Sub

Private Sub Raz_Click()
    ' Mise à zéro des équipes
    Eq1.Text = ""
    Eq2.Text = ""
    ' Mise à zéro des scores
    Score1 = 0
    Score2 = 0
    ' Mise à zéro des listes
    List1.Clear
    List2.Clear
End Sub

Private Sub Res_Click()
    ' Déclaration message
    Dim Message As String
    ' Message = Equipe 1 : Score - Equipe 2 : Score
    Message = Eq1.Text & " : " & CStr(Score1) & " - " & Eq2.Text & " : " & CStr(Score2)
    ' Appel à une boîte de message standard
    Call MsgBox(Message, , "Score actuel")
End Sub
```